

1 **Didactic Connection between Spreadsheet and Teaching** 2 **Programming**

3
4 *When we talk about problem-solving skills, then, generally, programming*
5 *comes to our minds as an activity that can develop algorithmic thinking and*
6 *abstraction. Regarding the spreadsheet, the software application area could*
7 *be our first, and mathematics could be our second thought. When*
8 *spreadsheets and programming are mentioned together, programming of*
9 *macros is in focus, which is in fact programming. In this paper, we want to*
10 *focus on how these two areas impact each other, and we want to emphasize*
11 *that the spreadsheet is an efficient tool to develop algorithmic thinking.*
12 *Moreover, there is more “crosstalk” between these two tools. This paper*
13 *will show through examples that there is a two-way connection between*
14 *spreadsheet and programming; that is why it can be useful to build the*
15 *concepts of these two topics mutually on each other.*

16
17 **Keywords:** *spreadsheet, programming, problem solving, algorithmic*
18 *thinking, teaching methodologies*

21 **Introduction**

22
23 Usually, spreadsheet teaching is not classified as a problem-solving tool.
24 For example, according to the Curriculum Framework of the National Core
25 Curriculum (NAT) 2012, the topic “Problem-solving with IT tools” deals with
26 only programming and algorithms (NAT, 2012). Spreadsheets are part of the
27 topic “Using application systems”, and the goal of spreadsheets is retrieving
28 information.

29 The new Curriculum Framework 2020 (NAT, 2020) adopts a different and
30 more suitable approach according to which spreadsheets are a new topic in the
31 field of “Developing problem-solving skills”. With this change of approach,
32 teaching methods developed earlier by the Faculty of Informatics of Eötvös
33 Loránd University, are followed (Zsakó, 2015a; Zsakó, 2015b). In our work,
34 we will show that the new Curriculum Framework has the right approach.

35 In our paper we will go through the key concepts of programming and we
36 will demonstrate how these concepts can be taught by spreadsheet. We will
37 focus, in particular, on how to present and teach programming theorems in
38 spreadsheets. For this reason, we will define the concept of programming
39 theorems after the literature review.

42 **Literature Review**

43
44 The spreadsheet teaching field includes basic programming concepts (Tort,
45 2010), like data types, operations, variables, and functions. Tort also suggests
46 adding procedures, scopes of variables, data tables, sorting, etc. because a
47 spreadsheet can be considered as a program and building a spreadsheet is partly

1 programming. If we use a model of a spreadsheet explicitly, then we can help
2 learners in the process of abstraction, which is a very important part of
3 programming.

4 According to Szalayné (2016), a table spreadsheet can be considered as a
5 program with data and pre-defined algorithms. Although students can see a
6 table/spreadsheet on their screens, they need to understand the “program”,
7 which consists of their solutions implemented by functions.

8 Csernoch and Bíró claim that spreadsheet software can be used as a
9 problem-solving tool. Their method, called Sprego, “*is a deep approach*
10 *metacognitive problem-solving environment, which has borrowed and*
11 *combined proven methods from high level programming languages. The three*
12 *milestones of Sprego are*

- 13
- 14 • *using as few and as simple general-purpose functions as possible,*
- 15 • *building multilevel formulas,*
- 16 • *building array formulas.”* (Csernoch & Bíró, 2015 p. 27)
- 17

18 This method can develop students’ computational thinking and algorithmic
19 skills. Teaching spreadsheet has an important role in ICT education because
20 students learn several aspects of computer science and develop skills connected
21 to this field, for example, handling data structures, database management,
22 programming principles, logical and computational thinking, and algorithmic
23 skills. Sprego also promotes schema construction through authentic problem-
24 solving and algorithm construction (Csapó, Csernoch & Abari, 2020).

25 Many fundamental programming concepts have their equivalents in
26 spreadsheet. Kankuzi et al. (2017) propose that before an introductory
27 programming course, students should learn spreadsheet programming, where
28 the fundamentals of programming are indirectly introduced to them through
29 problem solving by using spreadsheet.

30 According to Warren (2004), if we use spreadsheet before teaching a
31 programming language, then it takes less time to get to more complicated
32 algorithms.

33

34

35 **Programming Theorems/Patterns of Algorithms**

36

37 Programming tasks can be categorized into groups according to their
38 types, which is useful because for each group we can create an algorithm
39 rule/schema that solves all the tasks in that specific group. These task types are
40 called *programming theorems* because their solutions are justifiably the correct
41 solutions.

42 Essentially, programming theorems / patterns of algorithms are abstract
43 specifications and algorithms that we want to use as schemas in order to solve a
44 programming task. The aim of *specification* is to give the task in a formalized
45 way (it can be an “interface” between the programmer and the customer).
46 Specification has four components: input, output, precondition and

1 postcondition. *Input* is the input data of the task; *precondition* gives
 2 information on the input (i.e., which statements should be fulfilled by the input
 3 data); *output* is the result of the task; and *postcondition* is statements used to
 4 get the result (how we reach the result-state from the first input-state)
 5 (Harangozó et al., 1998).

6 We can recognize the suitable programming theorem from the task
 7 description. Once we have done this, we can use the specific data of the general
 8 task type, and in the general algorithm substitute them with the task-specific
 9 data. Applying this method will lead us to the correct solution.

10 In these tasks we usually have to assign a certain result to one (or more)
 11 data collection(s), which, for simplicity's sake, we will handle as some sort of
 12 sequences. In simple cases sequences can be illustrated as arrays
 13 (Szlávi et al., 2019).

14 Programming theorems / patterns of algorithms are proven templates as a
 15 basis on which we can build our solutions later. (This way development will be
 16 quicker and safer.) We note here that our term "Patterns of algorithms" differs
 17 from the usual definition (LMU). According to our wording, pattern refers to a
 18 task-schema and not to a problem-solving strategy.

19 We can categorize programming theorems in three groups. We would like
 20 to summarize the essence of these algorithms (Szlávi et al., 2019).

21 22 **Basic Programming Theorems** 23

- 24 • **Sequential computing (sequence calculations):** We have an input
 25 sequence, and we have to calculate a single value from that. We will
 26 use the same operation on every element of the sequence.
- 27 • **Counting:** We have an input sequence, and we have to count how
 28 many of them have a given attribute.
- 29 • **Decision:** Let us determine if there is an item with a given attribute
 30 among the elements of an input sequence.
- 31 • **Selection (linear selection):** We have an input sequence, and we have
 32 to select an element which has a given attribute, assuming that at least
 33 one such element exists in the input sequence.
- 34 • **Search (linear search):** We have an input sequence, and we have to
 35 search for an element that has a given attribute, and we do not know
 36 whether such an element exists in the sequence. (Search is the
 37 construction of *decision* and *selection*.) Dijkstra calls this algorithm
 38 "Linear search theorem" (1976., pp. 105).
- 39 • **Maximum selection:** We have to pick/find the greatest (or smallest)
 40 value from the input sequence.

41 42 **Complex Programming Theorems** 43

- 44 • **Copy (calculation with a function):** We have an input sequence with
 45 N elements, and we have to assign N other elements to these. The type
 46 of assigned values can differ from the type of original values, but the

1 count (N) remains the same, as well as the order. In other words, we
 2 will use the same operation on each of the elements of the sequence, but
 3 the output will be a sequence.

- 4 • **Multiple item selection:** We have to list all elements from the input
 5 sequence which have a common attribute A.
- 6 • **Partitioning:** We have to list all elements from an input sequence
 7 which have a common attribute A, and then also list those ones not
 8 having attribute A. So, we “assign” all the elements of the input to one
 9 of the output sequences. (Of course, there can be more than two
 10 attributes.)
- 11 • **Intersection:** We have two sets as input (with elements of the same
 12 type), and we have to list all elements that are part of both sets. (This is
 13 the construction of *multiple item selection* and *decision*.)
- 14 • **Union:** We have two sets as input (with elements of the same type), and
 15 we have to list all elements that are included at least in one of the sets.
 16 (This is the construction of *copy*, *multiple item selection* and *decision*.)

17 **Constructed Programming Theorems**

- 20 • **Conditional copy:** We will calculate the same operation on each
 21 element of the sequence which have the given attribute and another
 22 operation on each element which does not have the given attribute.
 23 (This is the construction of *multiple item selection* and *copy*.)
- 24 • **Conditional summation:** Sum of elements with a certain attribute.
 25 (This is the construction of *multiple item selection* and *summation*.)
- 26 • **Conditional maximum search:** find the maximum of the elements that
 27 have a certain attribute. (This is the construction of *decision* and
 28 *maximum selection*.)
- 29 • There are at least K elements with the given attribute (This is the
 30 construction of *search* and *counting*.)

31 **Basic Programming Concepts in Spreadsheet**


32 **Variable and Data Type**

33
 34
 35
 36
 37 Cells, one of the most important basic concepts of spreadsheets, are
 38 comparable with variables, one of the most important basic concepts of
 39 programming. Like variables, cells are named containers (they have a default
 40 name, but can also be renamed) in which data can be written and from which
 41 the same data can be retrieved. If the user enters the data, it corresponds to
 42 reading a value from the user into a variable. If a formula enters the data, it
 43 corresponds to storing the result of a calculation in a variable. However, unlike
 44 variables, cell content is constantly visible, so no instruction for displaying
 45 output is required. It is important to note that in spreadsheet it is not the cell
 46 that has a type, but the value stored in it, as even values from different types

1 can be written in the same cell. However, data validation can be set to a cell to
 2 limit the type of data that can be entered in it, which is like declaring the type
 3 of a variable.

4 An example of the specialty of spreadsheet’s variable concept is that we
 5 can assign a name to separated ranges as well and we can use it as a parameter
 6 (see Figure 1); we cannot do this in programming.

7
 8 *Figure 1.* Separated Ranges as a Single Variable and as a Parameter in
 9 Spreadsheet



10
 11
 12
 13
 14
 15
 16
 17
 18
 19
 20

The first column of Table 1 contains the “data types” of spreadsheet, while the second one shows the construction of these types in programming from primitive data types. “Data types” are enclosed in quotation marks in the first case because the spreadsheet does not implement them as true data types. We add that professional programming languages often include the appropriate composite data types so that the programmer does not have to construct them.

Table 1. “Data Types” in Spreadsheet and Their Construction in Programming

“Data Type” in Spreadsheet	Construction in Programming
Number	integer, real
Currency	integer, real + output formatting
Accounting	integer, real + output formatting
Date	integer, real + output formatting or record type (struct)
Time	integer, real + output formatting or record type (struct)
Percentage	integer, real + output formatting
Fraction	record type (struct) + output formatting
Scientific	integer, real + output formatting
Text	string
Logical	Boolean

21
 22
 23
 24

Behind the scenes, in fact, spreadsheet deals with 4 data types: logical, number, text and error (error data type is not the subject of our paper). All the “data types” of spreadsheet are real numbers except text and logical. This

1 means that all numeric “data types” of spreadsheet are representations; more
2 precisely, they are output formatting (see Table 2).

3
4 *Table 2. Representations of Spreadsheet’s Number Type*

“Data Type”	Displayed in Cell	Stored Number
Number	123456,00	123456
Currency	\$123 456,00	123456
Accounting	\$ 123 456,00	123456
Date	01.03.2238	123456
Time	12:00:00	123456,5
Date and Time	01.03.2238 12:00 PM	123456,5
Percentage	75,00%	0,75
Fraction	2/3	0,66666667
Scientific	1,23E+05	123456

5
6 There is a great similarity between the spreadsheet’s cell format and
7 programming languages’ formatted output (i.e., decimal places, format
8 numbers in thousands, etc.).

9 10 **Function and Data Type**

11
12 Understanding spreadsheets requires a function-like way of thinking
13 (introduction to functional programming). Using parametrizing functions and
14 nested functions in spreadsheet can support the understanding of parametrizing
15 and parameter passing in conventional programming languages.

16 In order to form the correct type-concept, spreadsheet has an important
17 role because there are specific functions that can be interpreted only on specific
18 types. For example, SUM and AVERAGE functions can be interpreted only on
19 numeric data, and each arguments of the logical functions, such as AND or
20 OR, must be logical values. Students can understand that the type is not only a
21 set but the applicable operations as well. There is a great difference between
22 digits as string and numbers (difference between “23” and 23). Constant data
23 show this difference as well.

24 25 **Array and Matrix**

26
27 Although spreadsheet has a special variable concept (Szlávi,
28 Törley & Zsakó, 2018), a deeper understanding of functions can support
29 students’ understanding of the difference between scalar and sequence and
30 what it means to travers a sequence. In spreadsheet, a sequence can be stored in
31 an array or in a matrix. The best tool for comprehending the concept of
32 indexing can be the INDEX function that executes the indexing operation on a
33 selected range.

34 A single cell is not suitable for storing complex types, such as a record, but
35 using several adjacent cells can be a solution.

1 **Record**

2
3 If we view a table in spreadsheet as a table in a database, then its rows can
4 be considered records, the fields of which are defined by the columns. That is,
5 the table can be considered an array of records, or even an array of objects,
6 which can lead to the concept of object-oriented programming.

7 8 **Conditional and Loop**

9
10 The IF function can help to understand the conditional control structure as
11 well as the logical (Boolean) type and operations (AND, OR functions).

12 Loop, as a language element, is not part of spreadsheets, but its concept
13 can be discovered on different levels. For example, if we perform the same
14 operation on all elements of a column in an adjacent column using a copied
15 formula (for example, calculating prices increased by some percentage), we are
16 processing the elements of the column just as a loop traverses an array. In
17 addition, elements of columns (or ranges) can be traversed using array
18 formulas.

19 Deeper comprehension of lookup functions can lead to the concept of
20 conditional loops because if we look for something, then we can pose a
21 question whether we need to examine all of the elements of the sequence in
22 order to give a definite answer.

23 24 25 **Programming Theorems in Spreadsheet**

26
27 Programming theorems can be demonstrated in three different levels in
28 spreadsheet:

- 29
30 1. with the appropriate built-in functions, students can become familiar
31 with the concept of programming theorems;
- 32 2. using spreadsheet as an algorithm visualization tool, students can
33 understand how programming theorems work;
- 34 3. most programming theorems can be implemented using array formulas
35 based on the postconditions of their specifications.

36 In the following, we would like to present these three levels.

37 38 **Understanding Programming Theorems Using Built-in Functions**

39
40 Problems to solve with spreadsheet and with programming are often
41 similar, so it is no surprise that the spreadsheet has the functions that
42 implement most of the programming theorems. Table 3 summarizes the
43 connections between spreadsheets and patterns of algorithms.

44
45

1 *Table 3. Connection between Spreadsheet and Patterns of Algorithms*

Patterns of Algorithms	Built-in Functions in Spreadsheet
sequential computing (conditional as well)	SUM, SUMIF, SUMIFS, AVERAGE, AVERAGEIF, AVERAGEIFS, DSUM, DAVERAGE, CONCAT
counting	COUNTIF, COUNTIFS, DCOUNT, DCOUNTA
decision	IF(COUNTIF), IF(COUNTIFS)
selection	VLOOKUP, HLOOKUP, XLOOKUP, INDEX(MATCH), DGET
search	<i>decision + selection</i>
maximum selection	MAX, MIN
copy (map)	there is not any special function, it can be implemented by copying the reference/formula (Figure 6.) or by creating an array formula
multiple item selection	filter and advanced filter
conditional maximum	MAXIFS, MINIFS, DMAX
K th maximum	LARGE, SMALL
Sort	SORT (sorting criteria exists but we do not know anything about the method)

2

3

4

5

6

7

8

9

10

11

12

13

14

15

16

17

18

19

20

21

22

23

24

25

26

It should be noted that the *selection* programming theorem can only be implemented in spreadsheet with crucial limitations. In the case of this programming theorem, the attribute to be examined can be any logical condition. On the contrary, lookup functions (except DGET) can only find an item equal to a specified value in an arbitrary (unordered) range. Although the DGET function can search using any logical condition, it only provides a solution if exactly one element meets the condition.

We would like to highlight *decision*, *selection* and *search* algorithms, showing how they can be implemented in spreadsheet. It can be presented that VLOOKUP and MATCH functions implement only the *selection* algorithm because they do not give any meaningful answer if the element which we looked for does not exist. *Decision* algorithm should be rephrased: Does the specific element or the element with the specific attribute exist? This way of thinking is connected to the postcondition of *decision* algorithm. As we deduct *linear search* algorithm from the construction of *decision* and *selection* algorithms, we will use the same construction in spreadsheet. For example:

$$\text{IF}(\text{COUNTIF}()>0;\text{VLOOKUP}();\text{"None"})$$

Algorithm Visualization of Programming Theorems

As mentioned above, spreadsheet has the functions with which most of the programming theorems can be implemented. However, these functions hide the actual calculations from the user. Spreadsheet can also support understanding

1 an algorithm step by step and in this way it can support understanding how an
 2 algorithm, such as a programming theorem, works. In other words, spreadsheet
 3 can visualize the input, the output and the state of the output variable at each
 4 step of the algorithm. This means that spreadsheet can show us the whole state
 5 space (i.e., input, output, local variables). As examples, we would like to
 6 present a possible visualization of the following programming theorems:
 7 *counting*, *maximum selection*, *decision*, *conditional maximum search*, and
 8 *copy*.

9 The *counting* programming theorem stores the current number of
 10 elements having a given attribute A in an auxiliary variable. It first sets the
 11 auxiliary variable to 0, then uses a For loop to traverse the sequence, and if the
 12 current element has attribute A, it increments the value of the auxiliary variable
 13 by 1.

14 In our example (see Figure 2), the sequence has 10 elements in an array,
 15 and the attribute A is whether the element is greater than 5. Our visualization
 16 shows the current value of the auxiliary variable in column Count using the
 17 formula shown in the figure.
 18

19 **Figure 2. Visualization of Counting Programming Theorem**

	A	B	C	D	E
1	i	X[i]		Count	N=10 A(X[i]) → X[i]>5 Counting(N, X, Count): Count:=0 For i:=1 to N do If A(X[i]) then Count:=Count+1 End For End.
2				0 initial value	
3	1	3		0 =IF(B3>5;D2+1;D2)	
4	2	7		1 =IF(B4>5;D3+1;D3)	
5	3	5		1 =IF(B5>5;D4+1;D4)	
6	4	6		2 =IF(B6>5;D5+1;D5)	
7	5	4		2 =IF(B7>5;D6+1;D6)	
8	6	8		3 =IF(B8>5;D7+1;D7)	
9	7	9		4 =IF(B9>5;D8+1;D8)	
10	8	8		5 =IF(B10>5;D9+1;D9)	
11	9	1		5 =IF(B11>5;D10+1;D10)	
12	10	4		5 =IF(B12>5;D11+1;D11)	

20
 21
 22 Figure 3 shows how to visualize the *maximum selection* programming
 23 theorem. This algorithm uses a For loop and checks whether the current value
 24 of the sequence (in our example: the array) is higher than the local maximum.
 25 If yes, then we change the value of variable MaxVal to the current value of the
 26 array. In the first step, the local maximum is the first element of the array and
 27 that is why we start the loop counter from 2.
 28

1 **Figure 3. Visualization of Maximum Selection Programming Theorem**

	A	B	C	D	E
1	i	X[i]		MaxVal	N=10
2	1	3		3	=B2
3	2	7		7	=IF(B3>D2;B3;D2)
4	3	5		7	=IF(B4>D3;B4;D3)
5	4	6		7	=IF(B5>D4;B5;D4)
6	5	4		7	=IF(B6>D5;B6;D5)
7	6	8		8	=IF(B7>D6;B7;D6)
8	7	9		9	=IF(B8>D7;B8;D7)
9	8	8		9	=IF(B9>D8;B9;D8)
10	9	1		9	=IF(B10>D9;B10;D9)
11	10	4		9	=IF(B11>D10;B11;D10)

```

Maximum(N, X, MaxVal):
  MaxVal:=X[1]
  For i:=2 to N do
    If X[i] > MaxVal then
      MaxVal:=X[i]
  End For
End.
    
```

2
3

4 The *decision* algorithm checks the elements of the array until attribute A
5 becomes true for the current element. In our example attribute A is that the
6 value is even. Since we do not need to always check all the elements of the
7 array, there is a while loop in the algorithm. In the while loop, we check
8 whether the current element has attribute A and then we increment variable i,
9 which means we go to the next element. If there are elements to be checked
10 and the current element did not have attribute A, we go into the loop, otherwise
11 we exit from the loop.

12 The visualization in spreadsheet in Figure 4 shows well that as soon as the
13 current element has Attribute A, the variable Exists changes from false to true
14 and after that it will not change back to false (if there is not any element with
15 attribute A then Exists will remain false). According to the algorithm, however,
16 if Exists is true then there is no need to check further. To emphasize this, we
17 can easily create a conditional formatting that darkens (or even hides) the cells
18 belonging to the skipped steps. In our example, conditional formatting was
19 applied to range \$D\$3:\$D\$12 with rule “=D2”.

20
21

Figure 4. Visualization of Decision Programming Theorem

	A	B	C	D	E
1	i	X[i]		Exists	N=10
2				FALSE	initial value
3	1	3		FALSE	=OR(D2;MOD(B3;2)=0)
4	2	7		FALSE	=OR(D3;MOD(B4;2)=0)
5	3	5		FALSE	=OR(D4;MOD(B5;2)=0)
6	4	6		TRUE	=OR(D5;MOD(B6;2)=0)
7	5	4		TRUE	=OR(D6;MOD(B7;2)=0)
8	6	8		TRUE	=OR(D7;MOD(B8;2)=0)
9	7	9		TRUE	=OR(D8;MOD(B9;2)=0)
10	8	8		TRUE	=OR(D9;MOD(B10;2)=0)
11	9	1		TRUE	=OR(D10;MOD(B11;2)=0)
12	10	4		TRUE	=OR(D11;MOD(B12;2)=0)

```

A(X[i]) → X[i] is even

Decision(N, X, Exists):
  Exists:=False
  i:=1
  While i≤N and not Exists
    Exists:=A(X[i])
    i:=i+1
  End While
End.
    
```

22
23

24 The *conditional maximum search* programming theorem searches for the
25 largest item in the series that satisfies the specified condition. Of course, it is
26 not certain that there is an element in the series that satisfies this condition,
27 which is why the output will also contain a logical value (variable Exists) that
28 will be true if and only if the condition was true for at least one element.

1 This programming theorem is based on the *maximum selection*
 2 programming theorem described above. Now, however, it is not certain that the
 3 first element can be considered the maximum so far; instead, minus infinity
 4 will be the initial value of the conditional maximum (variable CMax).
 5 Furthermore, the current maximum value is substituted with a larger element
 6 only if that larger element satisfies the condition. At the end, the output logical
 7 value is set to true if and only if the value of the conditional maximum differs
 8 from minus infinity (see Figure 5).

9
 10 *Figure 5. Visualization of Conditional Maximum Search Programming Theorem*

	A	B	C	D	E
1	i	X[i]		CMax	
2				-1E+99	initial value
3	1	3		-1E+99	=IF(AND(MOD(B3;2)=0;B3>D2);B3;D2)
4	2	7		-1E+99	=IF(AND(MOD(B4;2)=0;B4>D3);B4;D3)
5	3	5		-1E+99	=IF(AND(MOD(B5;2)=0;B5>D4);B5;D4)
6	4	6		6	=IF(AND(MOD(B6;2)=0;B6>D5);B6;D5)
7	5	4		6	=IF(AND(MOD(B7;2)=0;B7>D6);B7;D6)
8	6	8		8	=IF(AND(MOD(B8;2)=0;B8>D7);B8;D7)
9	7	9		8	=IF(AND(MOD(B9;2)=0;B9>D8);B9;D8)
10	8	8		8	=IF(AND(MOD(B10;2)=0;B10>D9);B10;D9)
11	9	1		8	=IF(AND(MOD(B11;2)=0;B11>D10);B11;D10)
12	10	4		8	=IF(AND(MOD(B12;2)=0;B12>D11);B12;D11)
13					
14				Exists	
15				TRUE	=D12<>-1E+99
16					
17					N=10
18					A(X[i]) → X[i] is even
19					Maximum(N, X, Exists, CMax) :
20					CMax:=-∞
21					For i:=1 to N do
22					If A(X[i]) and X[i] > CMax then
23					CMax:=X[i]
24					End For
25					Exists:=CMax≠-∞
26					End.

11
 12
 13 There is not any function that can directly implement the *copy (map)*
 14 algorithm. If we execute the same operation on the elements of the input
 15 sequence, the output will be a sequence. The “copying formula” (actually
 16 copying reference) feature of the spreadsheet shows that during the *copy*
 17 algorithm we “copy” the formula so we “copy” the operation as well. This way
 18 we can visualize the *copy* programming theorem (see Figure 6).

19
 20

1 **Figure 6. Visualization of Copy Programming Theorem**

	A	B	C	D	E
1	i	X[i]		Y[i]	
2	1	3		6 =2*B2	N=10 f(X[i]) → 2*X[i]
3	2	7		14 =2*B3	Copy(N, X, Y): For i:=1 to N do Y[i]:=f(X[i]) End For
4	3	5		10 =2*B4	
5	4	6		12 =2*B5	
6	5	4		8 =2*B6	
7	6	8		16 =2*B7	
8	7	9		18 =2*B8	
9	8	8		16 =2*B9	
10	9	1		2 =2*B10	
11	10	4		8 =2*B11	

2

3

4

Implementation of Programming Theorems Based on Their Postconditions

5

6

In the case of advanced spreadsheets, array formulas can map all the patterns of algorithms (programming theorems), and there can be a connection among array formulas and postconditions of programming theorems.

8

9

10

11

12

13

14

To understand the postcondition of some programming theorems, spreadsheet can be a good support. We need to use array formulas. In many cases, the implemented solution by spreadsheet is obvious: for example, *summation*, *counting*, *conditional summation*, *copy*, and *conditional copy*. For instance, the postcondition of *counting* looks like this:

$$Count := \sum_{i=1}^N 1_{A(Array_i)}$$

15

16

17

18

19

20

21

22

23

24

25

26

27

Where A is the attribute function, N is the size of the sequence (in this case: array). This means if the given array-element has attribute A then we add 1 to Count. In spreadsheet, this formula can be implemented literally with the array formula. The Greek letter great sigma means that we add more elements to each other and the condition below that decides at which elements we should add 1 to Count. The operation of great sigma will implement the SUM function, and the operation of the conditional will implement the IF function. That is why the following spreadsheet formula will implement the postcondition of count algorithm correctly:

$$\{=SUM(IF(A(array);1;0)\}$$

28

29

30

31

32

33

34

The SUM function will sum an array with elements 0 and 1 (the output of IF function) and those elements will be 1 that has A attribute (in other words: where the value of A function is true).

In our previous work (Szlávi, Törley & Zsakó, 2019), we have proven that all the programming theorems can be deduced to the *sequential computing* theorem. We have claimed that the *decision* algorithm deduced to *sequential computing* gives the correct solution based upon a Boolean array where the i^{th}

1 element of the array is true if the i^{th} element of the input array has A attribute.
 2 *Decision* algorithms have two variants: the first one checks if there is an
 3 element in the input array that has attribute A, while the second one checks if
 4 every element in the input array has attribute A. It can be proven easily that the
 5 following array formulas implement the *decision* algorithm:

6

- 7 • existing element with A attribute: {=OR(A(array_element))}
- 8 • every element with A attribute: {=AND(A(array_element))}

9

10 We note here that we could implement this theorem with “normal” (i.e. not
 11 array) formulas (for example COUNTIF, COUNTIFS functions) but this way
 12 of thinking would not lead us to an efficient algorithm and we could not
 13 connect it to the postcondition.

14 Array formulas could help to understand the combination/construction of
 15 programming theorems. A good example of this is the *conditional maximum*
 16 *search* algorithm that is the construction of *decision* and *maximum search*. We
 17 will combine the postcondition of these algorithms, which means if an element
 18 exists that has attribute A in the array then we calculate the maximum of these
 19 elements:

20

21 {=IF(OR(A(array_element));
 22 MAX(IF(A(array_element);array_element; “”);“NONE”)}
 23

24 The connection between algorithm patterns’ postconditions and array
 25 formulas can be seen in Table 4.

1 *Table 4.* The Connection of Algorithm Patterns' Postcondition and Array
 2 Formulas

Pattern of Algorithm and Postcondition	Array Formula Implementation
Summation (sequential computing) $\sum_{i=1}^N Array_i$	{=SUM(array)}
Counting $\sum_{i=1}^N 1$ $A(Array_i)$	{=SUM(IF(A(array);1;0))}
Decision (exists) $\text{exist} := \exists i \in [1..N]: A(Array_i)$	{=OR(A(array))}
Decision (all) $\text{all} := \forall i \in [1..N]: A(Array_i)$	{=AND(A(array))}
Conditional sum $\sum_{i=1}^N Array_i$ $A(Array_i)$	{=SUM(IF(A(array); array;0))}
Conditional maximum $\text{exist} := \exists i \in [1..N]: A(Array_i) \text{ and } \text{exist} \rightarrow \text{MaxVal}$ $= \text{MAX}_{i=1}^N Array_i$ $A(Array_i)$	{=IF(OR(A(array)); MAX(IF(A(array); array; ""));"NONE")}
Copy $\forall i \in [1..N]: F(Array_i)$	{=F(array)}
Conditional copy $\forall i \in [1..N]: A(Array_i) \text{ is true: } F(Array_i) \text{ else } Array_i$	{=IF(A(array);F(array); array)}
Multiple item selection $\text{Count} := \sum_{i=1}^N 1 \text{ and}$ $A(Array_i)$ $\forall i \in [1.. \text{Count}]: A(Y_i)$	{=IF(A(array); array; "")}

3
 4 We can see a connection between the formulas of postconditions and the
 5 formulas of spreadsheet. This can be seen on Table 5.

6
 7 *Table 5.* The Connection between Formulas in Postcondition and Formulas in
 8 Spreadsheet

Formula in Postcondition	Formula in Spreadsheet
$\sum_{i=1}^N Array_i$	SUM(array)
$MAX_{i=1}^N Array_i$	MAX(array)
F(Array _i)	F(array)
A(Array _i)	IF(A(array);array; "")
$\hat{S}i[1..N]: A(Array_i)$	OR(A(array))
$\hat{i}[1..N]: A(Array_i)$	AND(A(array))

1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16
17
18
19
20
21
22

Table 5 shows that we have “building blocks” and by combining these “blocks” a more complex postcondition can be built. This combination shows how programming theorems can be constructed.

We should take a note on *maximum selection* and *multiple item selection* programming theorems. *Maximum selection* cannot be implemented with an array formula because we cannot compare and refer to the elements of the array in the memory (like we showed in Figure 3). CMax can be implemented with an array formula because IF function can select those array elements for MAX function which have A attribute.

The result of *multiple item selection* is an array (which is Y in the postcondition). Count will be the number of those elements which have A attribute (like at count programming theorem) and it will be the number of elements of the output array in Figure 8. In spreadsheet, we do not need to output the number of the output array.

The implementation of some programming theorems with scalar output using array formulas, based on the postconditions of their specifications can be seen in Figure 7.

Figure 7. Implementation of Summation, Counting, Decision (in two variants), Conditional Summation and Conditional Maximum Search Programming theorems

	A	B	C	D	E	F	G
1	i	X[i]			Attribute	Solution	
2	1	3	Summation		—	55	{=SUM(B2:B11)}
3	2	7	Counting	X[i]>5		5	{=SUM(IF(B2:B11>5;1;0))}
4	3	5	Decision (exists)	X[i] is even	TRUE		{=OR(MOD(B2:B11;2)=0)}
5	4	6	Decision (all)	X[i] is even	FALSE		{=AND(MOD(B2:B11;2)=0)}
6	5	4	Conditional summation	X[i]>5		38	{=SUM(IF(B2:B11>5;B2:B11;""))}
7	6	8					
8	7	9	Conditional maximum search	X[i] is even		8	{=IF(OR(MOD(B2:B11;2)=0); MAX(IF(MOD(B2:B11;2)=0;B2:B11;"")); "NONE")}
9	8	8					
10	9	1					
11	10	4					

23
24
25
26
27
28

Similarly, Figure 8 shows the implementation of some programming theorems with array output. Due to the particularity of spreadsheet, the continuance of array cannot be kept at *multiple item selection* algorithm.

1 *Figure 8.* Implementation of Copy, Conditional Copy, and Multiple Item
 2 Selection Programming Theorems

	A	B	C	D	E	F	G	H
1	i	X[i]		Y[i]	Copy		Y[i]	Conditional copy
2	1	3		6	{=2*B2:B11}		3	{=IF(B2:B11>5;B2:B11*2;B2:B11)}
3	2	7		14			14	
4	3	5		10			5	
5	4	6		12			12	
6	5	4		8			4	
7	6	8		16			16	
8	7	9		18			18	
9	8	8		16			16	
10	9	1		2			1	
11	10	4		8			4	
12								
13								
14				Y[i]	Multiple item selection			
15					{=IF(B2:B11>5;B2:B11;"")}			
16					7			
17								
18					6			
19								
20					8			
21					9			
22					8			
23								
24								

3
 4
 5
 6 **Summary of the Three Levels through an Example**

7
 8 As stated earlier, programming theorems can be demonstrated at three
 9 different levels in spreadsheet. The first one is about the comprehension and
 10 usage of programming theorems using the proper built-in functions. The
 11 second one visualizes the algorithms of the programming theorems using only
 12 basic operators and functions. In the third level we can implement most of the
 13 programming theorems using array formulas, according to their specifications,
 14 or more precisely, postconditions. Consequently, all levels can help learning
 15 programming theorems from a different aspect.

16 For comparison, Figure 9 shows the appearance of the *counting*
 17 programming theorem at the mentioned three levels.
 18

1 *Figure 9.* Appearance of the Counting Programming Theorem at the Three
 2 Levels (The Solution Has a Thick Outside Border in Each Level).

	A	B	C	D	E	F	G	H	I
1	i	X[i]							
2	1	3							
3	2	7							
4	3	5							
5	4	6							
6	5	4							
7	6	8							
8	7	9							
9	8	8							
10	9	1							
11	10	4							
12									
13	Built-in function		Algorithm visualization			Implementation			
14	5	=COUNTIF(B2:B11;">5")	Count			5	={SUM(IF(B2:B11>5;1;0))}		
15			0	initial value					
16			0	=IF(B2>5;E15+1;E15)					
17			1	=IF(B3>5;E16+1;E16)					
18			1	=IF(B4>5;E17+1;E17)					
19			2	=IF(B5>5;E18+1;E18)					
20			2	=IF(B6>5;E19+1;E19)					
21			3	=IF(B7>5;E20+1;E20)					
22			4	=IF(B8>5;E21+1;E21)					
23			5	=IF(B9>5;E22+1;E22)					
24			5	=IF(B10>5;E23+1;E23)					
25			5	=IF(B11>5;E24+1;E24)					

Conclusions

Our paper showed why the spreadsheet (except table formatting and graphs) is part of computational thinking (together with algorithm and programming) rather than digital literacy. Spreadsheets and algorithms both involve problem-solving (skills).

We can find a great similarity between the topics (data, patterns and algorithms) of the two areas and that is why they can support each other when teaching students to learn and understand key concepts.

In the classical order of IT education, students learn spreadsheet before programming. That is why programming knowledge could be built upon spreadsheet (NAT, 2012; NAT, 2020; Szalayné, 2016). In Hungary, array formulas are taught only in talent development in secondary schools (Molnár, 2014), that is why they will not be the part of the regular teaching order); nevertheless our article intended to show that they could be essential tools in programming education.

References

- 1
2
- 3 Csapó G, Csernoch M, Abari K (2020) Sprego: case study on the effectiveness of
4 teaching spreadsheet management with schema construction. *Educ Inf Technol*
5 25, 1585–1605. <https://doi.org/10.1007/s10639-019-10024-2>
- 6 Csernoch M, Biró P (2015) Sprego programming. *Sprego Programming, Spreadsheets*
7 *in Education (eJSiE): Vol. 8: Iss. 1, Article 4.* [https://sie.scholasticahq.com/artic](https://sie.scholasticahq.com/article/4638-sprego-programming)
8 [le/4638-sprego-programming](https://sie.scholasticahq.com/article/4638-sprego-programming) (Retrieved: 29.01.2021.)
- 9 Dijkstra E W (1976) *A discipline of programming*. Prentice-Hall, Inc., Englewood
10 Cliffs, New Jersey
- 11 Harangozó É, Szlávi P, Zsákó L (1996) Joining Programming Theorems a Practical
12 Approach to Program Building, *Annales Universitatis Scientiarum Budapestinensis.*
13 *Sectio Computatorica*, Budapest, Hungary
- 14 Kankuzi B, Isong B, Letlonkane L (2017) Using the Spreadsheet Paradigm to
15 Introduce Fundamental Concepts of Programming to Novices, *In Proceedings of*
16 *SACLA'17*, July 3–5, 2017, Magaliesburg, South Africa
- 17 Loyola Marymount University (LMU): Definition of “Algorithmic patterns”
18 <https://cs.lmu.edu/~ray/notes/algpatterns/> (retrieved: 15.03.2021.)
- 19 Molnár K (2014) Tehetséggondozás az informatikában – Táblázatkezelés [Talent
20 development in informatics – Spreadsheet] ELTE Faculty of Informatics, [http://](http://tehetseg.inf.elte.hu/tananyagok/tablazatkez/index.html)
21 tehetseg.inf.elte.hu/tananyagok/tablazatkez/index.html (in Hungarian) (retrieved:
22 22.01.2021.)
- 23 NAT (2012) National Core Curriculum Framework for informatics in Hungary 2012.
24 https://kerettanterv.oh.gov.hu/05_melleklet_5-12/5.2.21_informat_5-10.doc
25 (retrieved: 22.01.2021.) (in Hungarian)
- 26 NAT (2020) National Core Curriculum Framework in Hungary 2020. [https://www.okta](https://www.oktas.hu/koznevelas/kerettantervek/2020_nat)
27 [tas.hu/koznevelas/kerettantervek/2020_nat](https://www.oktas.hu/koznevelas/kerettantervek/2020_nat) (retrieved: 22.01.2021.) (in Hungarian)
- 28 Szalayné Tahy Zs (2016) How To Teach Programming Indirectly – Using Spreadsheet
29 Application. *Acta Didactica Napocensia* 9 (1), 15-22, ISSN 2065-1430
- 30 Szlávi P, Zsákó L, Törley G (2019). Programming Theorems Have the Same Origin.
31 *Central-European Journal of New Technologies in Research, Education and*
32 *Practice*, 1(1), 1-12. <https://doi.org/10.36427/CEJNTREP.1.1.380>
- 33 Szlávi P, Törley G, Zsákó L (2018) The most difficult notion of programming: The
34 variable, In E. Száta, A. Buda (eds.) *Education - Technology - Computer Science*
35 *in Building better future*, Radom, Poland, Wydawnictwo Uniwersytetu
36 Technologiczno-Humanistycznego w Radomiu, 108-118
- 37 Tort F (2010) Teaching Spreadsheets: Curriculum Design Principles. *ArXiv, abs/1009.*
38 *2787.* <https://arxiv.org/ftp/arxiv/papers/1009/1009.2787.pdf> (Retrieved: 29.01. 2021)
- 39 Warren P (2004) Learning to program: spreadsheets, scripting and HCI, In
40 *Proceedings of the Sixth Australasian Conference on Computing Education – vol.*
41 *30*, Darlinghurst, Australia, 327–333.
- 42 Zsákó L (2015a) Informatika Nemzeti Alaptanterv 2020. [National Core Curriculum
43 in informatics 2020.] In P. Szlávi, L. Zsákó (eds.) *INFODIDACT 2015.* (Zamárdi,
44 Magyarország, 11.26.2015.-11.27.2015.) Budapest: Webdidaktika Alapítvány,
45 Paper 1. (ISBN: 978-963-12-3892-1) (in Hungarian)
- 46 Zsákó L (2015b) Informatikai tantervelmélet? Diszciplínák tanítása – a tanítás
47 diszciplínái 1. Tanulmányok a tudós tanár-képzés műhelyeiből [Curriculum
48 theory in informatics? Teaching of disciplines – disciplines of teaching vol. 1.
49 Essays from the workshop of scientific teacher training], *ELTE Eötvös Kiadó*,
50 Budapest, Hungary 92-111. (ISBN 978-963-284-611-8) (In Hungarian)